
ABSTRACT 


The goal of this project was to develop a program that would automate the process of generating high resolution three dimensional building models from Light Detection and Ranging (LiDAR) Elevation data.  The program was developed in Microsoft Visual Basic (VB) and ArcObjects as a Dynamic Linked Library (DLL) for ArcScene 3D.  The program relies upon 2D footprints previously generated by the Advanced LiDAR Exploitation System (ALES), but requires no further user interaction or input.
INTRODUCTION
LiDAR data is being collected by the army to aid in campaign and mission planning.  Commercial ArcGIS software is capable of representing LiDAR data in three dimensions, however only an outline of buildings is visible and the true structure of the building is masked.  ALES is capable of generating 2D footprints of buildings which can then be extruded to the average height of building to give it a 3D appearance and a flat roof.  However, most buildings have more complex roof types and thus flat building models would be insufficient.  Extracting high resolution building models from LiDAR data would provide information about the urban environment that would be essential in the planning operations.  The LiDAR Modeling System (LMS) developed by the University of Southern California (USC) is capable of extracting the entire 3D geometry of a building, but requires the user to select the roof type and points that define the specified roof type.  Thus, it is slow and time consuming and not feasible for rapidly modeling an urban area.  Software is needed that will automate the process of extracting building models with the need of little to no user interaction.

BACKGROUND

The army deploys aerial laser scanning to quickly collect data about an urban environment that could be used to help in mission planning.  LiDAR data was collected over several different areas, including San Francisco, Baltimore, Washington D.C Mall, Fort Benning, and the Engineering Research and Development Center’s (ERDC) Topographical Engineering Center (TEC).  The LiDAR data is converted to a raster with elevation values for every meter accurate to within 10 centimeters.
The raster can be loaded into ESRI ArcGIS software such as ArcMap and ArcScene.  Both ArcMap and ArcScene are capable of classifying the elevation data on a grayscale and provide a way of visualizing the raster data.  Additionally, ArcScene is capable of visualizing the data in three dimensions by extruding the raster values.  ArcObjects can be used to develop custom tools that take advantage of the built in features and capabilities of ArcScene.
PROCEDURE


A program had to be designed and developed to implement an algorithm for generating building models.  Instead of making a program that ran over the entire raster dataset, it would be more efficient to run for each building.  Before this could be done, individual buildings had to be isolated from the raster for future processing.  This required another program that could intelligently and accurately recognize and eliminate extraneous terrain data.  Instead of creating a new program for this specific task, the 2D footprints generated by ALES were adapted to this task.  Using 2D footprints to isolate one building at a time lowered the memory requirements and decreased the processing time required for modeling buildings.

 The LMS software that is capable of modeling buildings with user interaction requires two inputs from the user.  First it requires the user to recognize and select the type of roof that will be modeled, and then requires the user to select a few key points that could be used to generate the model.  The number and type of key points required depends on the roof type.  For example, modeling a sphere would require two key points.  One point would define the center of the sphere; the other key point would be an arbitrary key point on the surface of the sphere and would in essence be providing information about the radius of the sphere.  With the center and the radius of the sphere, it would be fairly easy to model.  For a flat roof surface, as many points as there are edges in the polygon defining the roof are required to accurately create a model of the roof.  Since the program will automate this process, it must replace the user and be able to first recognize a particular roof type and then find the points that define that particular roof type.

Though only given elevation data, there are two sets of data values which are useful for this purpose.  They are elevation or height values, which are given, and slope values, which can be easily calculated.  In fact the 3D Analyst Toolbox in ArcScene provides a tool which is capable of calculating the maximum slope in any direction from a particular point, or for all the points in the raster dataset. 

Now that we have height and slope datasets easily accessible as well as the area which we wish to model, we can begin to implement an algorithm that will recognize the roof as a particular type and then find the key points to define the roof.


The roof types considered shall be generalized as being a flat roof, slanted roof, peaked roof, cylinder, tower, sphere, or ellipsoid.  A flat roof has a polygonal base and is extruded to the same height for each point of the base.  A slant roof has a rectangular base with one side higher than the opposite and the adjacent sides of equal height. A peaked roof is two slanted roofs that meet at their peeks, the roof typical of most houses.  A cylinder is a flat roof with a circular base.  A tower is similar to a peaked roof except more than two slanted roofs meet at one high point.  The sphere and the ellipsoid can be parts of their respective shape, and are usually so.  Many buildings, especially larger ones, employ various combinations of these roofs and therefore the shapes must be recognized and modeled separately.  Furthermore, a hole may exist within the roof, which adds another dimension to the problem.

The first approach used in modeling the buildings relied solely upon height data and did not take advantage of slope data available.  The first step in this process was to break up the region of interest into groups of continuous points with the same height.  Because roofs are not made perfectly and the measurements taken by the LiDAR instrument is only accurate to within 10cm, a height tolerance had be used to distinguish points of the same height.  If the two heights have a difference less than or equal to the height tolerance, then they can be considered to be of the same height and can be grouped together.

Next, for each group of points a convex hull is created.  This significantly reduces the number of points in the group and creates a simply polygon that ArcGIS is capable of drawing.  Once the convex hull has been created, an intersection is taken with the original area that was selected for modeling.  This is necessary because the data originally analyzed in order to find groups of points was larger than the area of interest.  This is because a clip created from the raster data using a 2D footprint or user input will be rectangular and lie along the coordinate axis of the raster.  However, the models don’t usually lie along the coordinate axis and don’t necessarily have a rectangular shape, thus the actual area being modeled is larger and encompassing of the building itself. (See Figure 1)  By taking the intersection of the convex polygon with the building the area outside the building but inside of the raster clip is eliminated. (See Figure 2) 

The convex polygons are indicative of the shell of a feature in the roof of the building, but don’t contain information about the inside of the roof and thus obscure the true form of the building.  To add details of the interior of the roof to the building model I added points to each convex hull.  For each vertex of the convex hull, the highest point that could be reached by taking the steepest path from that vertex was added to the convex hull.  ArcGIS uses these interior points to create a polygon that defines the interior shape of the polygon.  The result is a polygon donut, the interior points will create a hole in the convex hull we had before. (See Figure 3)

Despite the obvious drawback of having a hole in the model, the hole does model the interior of the building and increases the accuracy of the polygon model.  The first attempt to fill the hole was basically creating a new polygon with just the interior points that were added later on.  This worked, but not perfectly, for ArcGIS used a convex hull for the interior solid and thus the polygon did not completely fill the hole. (See Figure 4)

Upon inspection of the donut polygon, it became obvious that ArcGIS had modeled the shape as one polygon as an incomplete ring instead of two disjoint polygons – one for the outside and one for the inside.  ArcGIS has the capability of defining each polygon as a collection of rings – which is closer to the polygon taught in mathematics.  A polygon created from a collection of rings can have holes defined by exterior and interior rings.  Very simply, area inside an interior ring is considered a hole in the polygon, while area inside an exterior ring is part of the solid polygon.  So now the original convex hole was set as the exterior ring of one polygon, and a convex hull of the interior points was set as the interior ring of the same polygon.  This creates a true donut shape, but it more or less similar to the holed polygon from before.  Another polygon is created with the exterior polygon set to the convex hull of the interior points.  This effectively and exactly fills the hole.
The model now consisted of a completely solid shape that had both interior and exterior features.  There was still a problem, not all of the interior points previously discovered were being represented – only the ones that formed the interior convex hull were affecting the model.   Furthermore, the most interior points – most often the points that were also the highest and thus would improve our model the most – were the ones being left out.

To fix this, the process of creating a polygon from an exterior and interior ring was modified slightly to run as many times as possible.  First using the collection of points that is the vertices of the original convex hull and all the interior points, a convex hull is created and set as the exterior ring of a new polygon.  Then, these points are removed and a new convex hull is created and set as the interior ring of the same polygon.  These two steps are repeated on the remaining points an indefinite number of times, until there are so few points remaining that a convex hull can no longer be created.  This improves greatly upon the old process because it allows for as much detail as possible for the modeling of the interior roof of the building. (See Figure 5)

 There are several aspects of this technique that could be improved upon and the technique is far from being capable of accurately modeling all of the various types of building roofs.  It can’t recognize and model a hole in the center of any building, no matter of what shape.  It is also not able to model a sphere or ellipsoid realistically.  Another problem with the models generated is that many small shapes are added to each model along with the few major shapes actually necessary.  The smaller models add extrusions and fine detail but obscure and hinder the display of the more important major shapes.  These number of these smaller models can be reduced my removing shapes smaller than a certain area.  However, care must be taken to ensure the minimum required area is not set too high, for this will remove shapes that may be essential to a building model.
CONCLUSION

The process explained above is successful in modeling buildings to an extent.  It does correctly differentiate between several different roofs types in one building, and will model each of them separately.  Its major shortcomings come from its lack of ability to actually recognize the type of roof it is modeling.  Because of this the program is unable to refine models by snapping them to key points.  It is also unable to remove all extraneous shapes that are already modeled within larger shapes and are unnecessary detailed.  The benefit of this approach is that it is capable of creating models for roofs not originally considered and is more versatile than a program that would try to characterize a roof from a built-in set of roofs.

To improve upon this process slope values should be taken into consideration, and the program should attempt to recognize roof types.  This would be especially beneficial in modeling spheres since spheres can’t be easily created from a group of polygons.  It would also provide for the identification and modeling of holes, a roof feature currently ignored by the program.

Appendix A: User Guide

The Urban Modeler Toolkit DLL includes three tools that can be used in ArcScene.  To use these tools, first import the DLL and then add the tools to the toolbar.   

Generate 2D Footprints

The “Generate 2D Footprints” tool will become active once a raster layer has been imported and it is selected.  To use the tool, select the tool and then select four points.  The four points will serve to create a quadrilateral that will be used as a 2D footprint.  The area within the quadrilateral will be modeled.
Process 2D Blueprints
The “Process 2D Blueprints” tool will become active once a raster layer has been imported and it is selected.  For this tool, turn the visibility of all 2D Footprint layers on, and turn the visibility of the raster layer off.  Select the raster layer and run the tool by clicking on it.  Every feature in every visible shape file will be used as a 2D footprint and the enclosed area will be modeled using the DoStuff sub in the HeightUse Module.

Process Blueprints

The “Process Blueprints” tool will become active once a raster layer has been imported and it is selected.  For this tool, turn the visibility of all 2D Footprint layers on, and turn the visibility of the raster layer off.  Select the raster layer and run the tool by clicking on it.  Every feature in every visible shape file will be used as a 2D footprint and the enclosed area will be modeled using the SlopeModel sub in the SlopeUse Module.
Appendix B: Source Code
'Module Coordinates

'Contains Functions related to the mapping of points in different coordinate systems

'Screen, Geographic, and File Coordinates are supported

Option Explicit

Private pPCSystem As IProjectedCoordinateSystem

'GetFileCoords returns a Point representing the File Coordinates of the Point on the given raster

'It requires a raster and a point for input, because the file coordinate of the point is dependent upon the raster

Public Function GetFileCoords(pRaster As IRaster, pPoint As IPoint) As IPoint

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    Dim pEnvelope As IEnvelope

    Set pEnvelope = pRProps.Extent

    Dim pRAProps As IRasterAnalysisProps

    Set pRAProps = pRaster

    Dim dPWidth As Double

    Dim dPHeight As Double

    dPWidth = pRAProps.PixelWidth

    dPHeight = pRAProps.PixelHeight

    Set GetFileCoords = CreatePoint( _

        Int(Abs(pPoint.x - pEnvelope.XMin - (0.5 * dPWidth)) / dPWidth), _

        Int(Abs(pRProps.Height - (pPoint.y - pEnvelope.YMin - (0.5 * dPHeight))) / dPHeight))

'   For a yet uknown reason, the y values must be flipped to ensure correct orientation

'   Because of floating point arithmetic and integer casting,

'   sometimes the function will return invalid maximum values

'   To fix this, the maximum valid value is returned instead

    If GetFileCoords.x = pRProps.Width Then GetFileCoords.x = pRProps.Width - 1

    If GetFileCoords.y = pRProps.Height Then GetFileCoords.y = pRProps.Height - 1

End Function

'GetUMDCoords converts a screen coordinate to a geographic coordinate

'Screen coordinate is defined by the x and y location of the pixel clicked on

Public Function GetUMDCoords(x As Long, y As Long) As IPoint

    Dim pScene As IScene

    Set pScene = pSDocument.Scene

    Dim pSGraph As ISceneGraph

    Set pSGraph = pScene.SceneGraph

    Dim pUnknown As stdole.IUnknown

    pSGraph.Locate pSGraph.ActiveViewer, x, y, _

        esriScenePickAll, True, GetUMDCoords, pUnknown, pUnknown

End Function

'GetUDMCoords is the inverse function to GetFileCoords

'Given a raster and a file coordinate on the raster, the corresponding geographic coordinate is returned

Public Function GetUDMCoords(ByVal x As Integer, ByVal y As Integer, pRaster As IRaster) As IPoint

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    Dim pEnvelope As IEnvelope

    Set pEnvelope = pRProps.Extent

    Dim pRAProps As IRasterAnalysisProps

    Set pRAProps = pRaster

    Dim dPWidth, dPHeight As Double

    dPWidth = pRAProps.PixelWidth

    dPHeight = pRAProps.PixelHeight

    Set GetUDMCoords = CreatePoint( _

        Abs(x * dPWidth + (0.5 * dPWidth) + pEnvelope.XMin), _

        Abs((pRProps.Height - y) * dPHeight + (0.5 * dPHeight) + pEnvelope.YMin))

'   Once again, y values are inversed

'   The problem with the y values is not related to both functions inversing y values

'   If the inverse is removed from both functions, y values do not map correctly

End Function

'GetSpatialReference was intended to be used by the Layer Module, specifically the CreateLayer function

'It is in this module because it relates to the coordinate system more than to a layer

'However, it is not used because creating more than one layer with this spatial reference causes an error

'Since multiple layers are needed, the spatial reference is instead hand coded in the CreateLayer function

Public Function GetSpatialReference(pRLayer As IRasterLayer) As IProjectedCoordinateSystem

    If pPCSystem Is Nothing Then

        Dim pGDataset As IGeoDataset

        Set pGDataset = pRLayer

        Set pPCSystem = pGDataset.SpatialReference

    End If

End Function

'CreatePnt creates and returns a Pnt defined by the x and y values passed

'A Pnt is used for manipulating pixel blocks, so this function is called by only the Values Module

'All other needs are met by the CreatePoint function below

Public Function CreatePnt(ByVal x As Double, ByVal y As Double) As IPnt

    Set CreatePnt = New Pnt

    CreatePnt.SetCoords x, y

End Function

'CreatePoint creates and returns a point from the coordinate input

'The point is not ZAware and doesn't have a set Z value

'CreatePoint is used by many functions and modules

Public Function CreatePoint(ByVal x As Double, ByVal y As Double) As IPoint

    Set CreatePoint = New Point

    CreatePoint.x = x

    CreatePoint.y = y

End Function

'Module Global

'Handles the initialization and value of global variables

'Only variables that are constant throughout the executation of

'the program and are required by many modules are defind here.

'The Global Module checks before initialization of any variable

'because some methods may be called more than once during execution,

'but only one instance of the variable is needed.

Option Explicit

Public pApplication As IApplication

Public pSDocument As ISxDocument

Public pMPatchLayer As IFeatureLayer

Public pPointLayer As IFeatureLayer

Public pPolygonLayer As IFeatureLayer

Public pMPointLayer As IFeatureLayer

Public pPolylineLayer As IFeatureLayer

'See comment concerning AddTriangleLayer and AddTStripLayer below

'Public pTriangleLayer As IFeatureLayer

'Public pTStripLayer As IFeatureLayer

'Init is called by the _OnCreate method of every Class Module

'Init initializes pApplication and pSDocument

Public Sub Init(ByVal hook As Object)

    If pApplication Is Nothing Then Set pApplication = hook

    If pSDocument Is Nothing Then Set pSDocument = pApplication.Document

End Sub

'Methods for creating layers

'Before creation, layers are checked to see if they have already been added,

'for adding the same layer again would delete the old layer and anything added to it.

Public Sub AddMPatchLayer()

    If pMPatchLayer Is Nothing Then Set pMPatchLayer = AddLayer("MultiPatch", esriGeometryMultiPatch, True)

End Sub

Public Sub AddPointLayer()

    If pPointLayer Is Nothing Then Set pPointLayer = AddLayer("Point", esriGeometryPoint, False)

End Sub

Public Sub AddPolygonLayer()

    If pPolygonLayer Is Nothing Then Set pPolygonLayer = AddLayer("Polygon", esriGeometryPolygon, False)

End Sub

Public Sub AddMPointLayer()

    If pMPointLayer Is Nothing Then Set pMPointLayer = AddLayer("MultiPoint", esriGeometryMultipoint, False)

End Sub

Public Sub AddPolylineLayer()

    If pPolylineLayer Is Nothing Then Set pPolylineLayer = AddLayer("Polyline", esriGeometryPolyline, False)

End Sub

'Layers for TriangleStrip and Triangles cause a "Graphics Renderer" error, and were thus removed

'Public Sub AddTStripLayer()

'    If pTStripLayer Is Nothing Then Set pTStripLayer = AddLayer("Triangle Strip", esriGeometryTriangleStrip, True)

'End Sub

'

'Public Sub AddTriangleLayer()

'    If pTriangleLayer Is Nothing Then Set pTriangleLayer = AddLayer("Triangles", esriGeometryTriangles, True)

'End Sub

'Module Layer

'Contains functions related to adding layers to the scene and adding features to layers

Option Explicit

'AddFeature adds a feature to the passed FeatureLayer

'AddFeature doesn't check geometry type of layer and of passed geometry match,

'so they must be the same or an error will be thrown

'iPoints is an optional field used viewing some data about the feature

'Originally used to see how many points the feature contained

'May be used for other debugging and testing purposes, but not used in most cases

Public Sub AddFeature(pFLayer As IFeatureLayer, pGeometry As IGeometry, Optional ByVal iPoints As Integer = -1)

    Dim pFClass As IFeatureClass

    Set pFClass = pFLayer.FeatureClass

    Dim pFeature As IFeature

    Set pFeature = pFClass.CreateFeature

    Set pFeature.Shape = pGeometry

    If Not iPoints = -1 Then pFeature.Value(2) = iPoints

    pFeature.Store

End Sub

'AddLayer creates and adds a layer to the scene

'The name and type of the layer must be passed, as well as if the layer is ZAware

'If the layer is ZAware, then only ZAware objects can be added

Public Function AddLayer(sLName As String, eGType As esriGeometryType, bZ As Boolean) As IFeatureLayer

    Dim pWFactory As IWorkspaceFactory

    Set pWFactory = New ShapefileWorkspaceFactory

    Dim sRDir As String

    sRDir = "C:\Documents and Settings\u4trrnrk\My Documents\SEAP\Data"

    Dim sFolder As String

    sFolder = "workspace"

    Dim sPath As String

    sPath = sRDir & "\" & sFolder & "\" & sLName

'   ArcObjects throws an error when trying to create a shapefile that already exists

'   So if the shapefile being created already exists, the existing shapefile is deleted

    If Dir(sPath & ".shp") = sLName & ".shp" Then

        Kill sPath & ".*"

    End If

    Dim pWName As IWorkspaceName

    Set pWName = pWFactory.Create(sRDir, sFolder, Nothing, 0)

    Dim pName As IName

    Set pName = pWName

    Dim pFWorkspace As IFeatureWorkspace

    Set pFWorkspace = pName.Open

    Dim pSREnvironment As SpatialReferenceEnvironment

    Set pSREnvironment = New SpatialReferenceEnvironment

'   The Projected Coordinate System is hard coded, because retrieving

'   the Projected Coordinate System and then creating a layer only allows

'   for one layer to be created before errors are given. See Function

'   GetSpatialReference in Module Coordinates for more information

    Dim pPCSystem As IProjectedCoordinateSystem

    Set pPCSystem = pSREnvironment.CreateProjectedCoordinateSystem(esriSRProjCS_WGS1984UTM_16N)

    Dim pGDef As IGeometryDef

    Set pGDef = New GeometryDef

    Dim pGDEdit As IGeometryDefEdit

    Set pGDEdit = pGDef

    With pGDEdit

        .GeometryType = eGType

        .HasZ = bZ

        Set .SpatialReference = pPCSystem

    End With

    Dim pField As IField

    Set pField = New Field

    Dim pFEdit As IFieldEdit

    Set pFEdit = pField

    With pFEdit

        .Name = "shape"

        .Type = esriFieldTypeGeometry

        .AliasName = "geometry"

        Set .GeometryDef = pGDef

    End With

    Dim pFEdits As IFieldsEdit

    Set pFEdits = New Fields

    pFEdits.AddField pField

    Set pField = New Field

    Set pFEdit = pField

'   An additional field for debuggin/testing purposes,

'   originally used for displaying how many points the feature contained.

    With pFEdit

        .Name = "points"

        .Type = esriFieldTypeInteger

    End With

    pFEdits.AddField pField

    Dim pFields As IFields

    Set pFields = pFEdits

On Error GoTo yea:

    Dim pFClass As IFeatureClass

    Set pFClass = pFWorkspace.CreateFeatureClass(sLName, pFields, Nothing, Nothing, esriFTSimple, "shape", "")

    Dim pFLayer As IFeatureLayer

    Set pFLayer = New FeatureLayer

    Set pFLayer.FeatureClass = pFClass

    pFLayer.Name = sLName

'   If the Layer is added to the Scene after the features are added,

'   then the Layer doesn't have to be refreshed.  Perhaps a future change

'   will be to move the following line to a point in execution time after

'   all features have been added to the layer.

    pSDocument.Scene.AddLayer pFLayer

    Set AddLayer = pFLayer

    Exit Function

yea:

    MsgBox ERR.Number

End Function

'Module Raster

'Contains Functions dealing with rasters

Option Explicit

'GetRasterSurfac returns the RasterSurface os the selected raster

'The RasterSurface is used for some raster surface operations in other modules

Public Function GetRasterSurface() As IRasterSurface

    Dim pRLayer As IRasterLayer

    Set pRLayer = GetRasterLayer()

    Dim pLExtensions As ILayerExtensions

    Set pLExtensions = pRLayer

    Dim p3DProperties As I3DProperties

    Dim i As Integer

    For i = 0 To pLExtensions.ExtensionCount - 1

        If TypeOf pLExtensions.Extension(i) Is I3DProperties Then

            Set p3DProperties = pLExtensions.Extension(i)

            Exit For

        End If

    Next

    If p3DProperties Is Nothing Then

        Dim pRBCollection As IRasterBandCollection

        Set pRBCollection = pRLayer.Raster

        Set GetRasterSurface = New RasterSurface

        GetRasterSurface.RasterBand = pRBCollection.Item(0)

    Else

        Set GetRasterSurface = p3DProperties.BaseSurface

    End If

End Function

'GetRaster returns the raster from the raster layer

Public Function GetRaster() As IRaster

    Set GetRaster = GetRasterLayer().Raster

End Function

'GetRasterLayer returns the selected layer as a raster layer

Public Function GetRasterLayer() As IRasterLayer

    Dim pRLayer As IRasterLayer

    Set pRLayer = pSDocument.SelectedLayer

    Set GetRasterLayer = New RasterLayer

    GetRasterLayer.CreateFromFilePath pRLayer.FilePath

End Function

'Clip returns a raster clip

'The clip is created from the passed raster using the envelope passed

Public Function Clip(pRaster As IRaster, pEnvelope As IEnvelope) As IRaster

    Dim pRTOp As ITransformationOp

    Set pRTOp = New RasterTransformationOp

    Dim pRDataset As IRasterDataset

    Set pRDataset = pRTOp.Clip(pRaster, pEnvelope)

    Set Clip = pRDataset.CreateDefaultRaster

End Function

'Module Values

'Contains functions for getting various values from the raster

Option Explicit

'GetPixelArray returns a matrix of the values in the raster passed

'GetPixelArray uses a single pixel block that spans the entire raster

Public Function GetPixelArray(pRaster As IRaster) As Variant

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    Dim pBlock As IPixelBlock

    Set pBlock = pRaster.CreatePixelBlock(CreatePnt(pRProps.Width, pRProps.Height))

    pRaster.Read CreatePnt(0, 0), pBlock

    GetPixelArray = pBlock.SafeArray(0)

End Function

'GetValueBlock returns a matrix of the values within the envelope in the raster

Public Function GetValueBlock(pRaster As IRaster, pEnvelope As IEnvelope) As Variant

    Dim pBlock As IPixelBlock

    Set pBlock = pRaster.CreatePixelBlock(CreatePnt(pEnvelope.Width, pEnvelope.Height))

    pRaster.Read CreatePnt(pEnvelope.XMin, pEnvelope.YMin), pBlock

    GetValueBlock = pBlock.SafeArray(0)

End Function

'GetSlope returns a raster that contains the slope values of the raster passed

Public Function GetSlope(pRaster As IRaster) As IRaster

    Dim pRSOp As ISurfaceOp

    Set pRSOp = New RasterSurfaceOp

    Set GetSlope = pRSOp.Slope(pRaster, esriGeoAnalysisSlopeDegrees)

End Function

'GetAspect returns a raster that contains the aspect values of the raster passed

Public Function GetAspect(pRaster As IRaster) As IRaster

    Dim pRSOp As ISurfaceOp

    Set pRSOp = New RasterSurfaceOp

    Set GetAspect = pRSOp.Aspect(pRaster)

End Function

'GetPixelAspect returns the aspect value of the raster at the passed point

Public Function GetPixelAspect(pRaster As IRaster, pPoint As IPoint) As Double

    Dim pSurface As ISurface

    Set pSurface = GetRasterSurface()

    GetPixelAspect = pSurface.GetAspectDegrees(pPoint)

End Function

'GetPixelSlope returns the slope value of the raster at the passed point

Public Function GetPixelSlope(pRaster As IRaster, pPoint As IPoint) As Double

    Dim pSurface As ISurface

    Set pSurface = GetRasterSurface()

    GetPixelSlope = pSurface.GetSlopeDegrees(pPoint)

End Function

'GetPixelValue returns the value of the raster at the passed point

Public Function GetPixelValue(pRaster As IRaster, pPoint As IPoint) As Variant

    Dim pBlock As IPixelBlock

    Set pBlock = pRaster.CreatePixelBlock(CreatePnt(1, 1))

    pRaster.Read CreatePnt(pPoint.x, pPoint.y), pBlock

    GetPixelValue = pBlock.GetVal(0, 0, 0)

End Function

'GetHeights returns a matrix of the values in the raster

'GetHeights uses IRasterCursor instead of a single pixel block to load values

Public Function GetHeights(pRaster As IRaster) As Variant

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    Dim aHeights() As Variant

    ReDim aHeights(pRProps.Width, pRProps.Height)

    Dim pCursor As IRasterCursor

    Set pCursor = pRaster.CreateCursor

    Dim pBlock As IPixelBlock

    Dim pSArray As Variant

    Dim iOX, iOY As Integer

    Dim iX, iY As Integer

    Do

        Set pBlock = pCursor.PixelBlock

        'pSArray = pBlock.SafeArray(0)

        iOX = pCursor.TopLeft.x

        iOY = pCursor.TopLeft.y

        For iX = 0 To pBlock.Width - 1

            For iY = 0 To pBlock.Height - 1

                'aHeights(iOX + iX, iOY + iY) = pSArray(iX, iY)

                aHeights(iOX + iX, iOY + iY) = pBlock.GetVal(0, iX, iY)

            Next iY

        Next iX

    Loop While pCursor.Next

    GetHeights = aHeights

End Function

'Module HeightUse

'HeightUse contains various functions that are used to create models of buildings

'HeightUse only uses height values and doesn't rely upon slope, curves, or aspects

'HeightUse will not try to recognize the type of building it is modeling,

'it has a generalized algorithm for creating a model regardless of building shape

'This gives it the versatility to model many types of buildings, but doesn't allow

'it to refine the models and correct for Lidar inaccuracies

Option Explicit

'Arrays used by the module.  aSlope and aAspect are not actually used, but they are loaded

Private aHeight As Variant

Private aSlope As Variant

Private aAspect As Variant

Private aVisited As Variant

Private aAdded As Variant

'Arrays of Objects used by the module

'aQueue is a pseudo-queue of points

Private aQueue As esriSystem.IArray

Private aOPolygon As esriSystem.IArray

Private aIPolygon As esriSystem.IArray

'Total Height and Height Error

Private dTHeight As Double

Private dHError As Double

'Dimension of the raster - maximum width and height

Private iMWidth As Integer

Private iMHeight As Integer

Private ltmp As Long

'Interfaces that are used many times in the module

Private pPCollection As IPointCollection

Private pZAware As IZAware

Private pEVertex As IEnumVertex

Private pPoint As IPoint

'Globals for the module

Private pGRaster As IRaster

Private pGTOperator As ITopologicalOperator

Private pGROperator As IRelationalOperator

Private pGROperator2 As IRelationalOperator

'Collections of rings

Private aERing() As IRing

'Private aRArray() As IRing

Private aIRing() As IRing

'Complete_Polygon uses the points in pPCollection to create polygons that model the points

'pPCollection contains the points that will be modeled

Public Sub Complete_Polygon(ByVal dAHeight As Double)

Dim pPolygon As IPointCollection

Set pPolygon = New Polygon

pPolygon.SetPointCollection pPCollection

'Polygons with an area less than 10 are artifacts that can be ignored

Dim pArea As IArea

Set pArea = pPolygon

If pArea.Area < 10 Then Exit Sub

'This will create a convex hull from the points and set it as the exterior ring of a polygon

'From the remaining points (the points inside of the conves hull), another convex hull is created,

'and it is set as the interior ring of the polygon

'This creates a donut polygon -  a polygon with a hole

'The process is repeated on the remaining points until no more convex hulls can be made (too few points)

'Note that the interior ring of a polygon will be the exterior ring of the next polygon,

'thus the hole will be filled, and the polygon will look complete.

On Error GoTo CPError:

While True

    Dim p As Integer

    Dim pGCollection As IGeometryCollection

    Set pGCollection = New Polygon

    Dim pTOperator As ITopologicalOperator2

    Set pTOperator = pPCollection

    p = 1000

    Dim pCHPCollection As IPointCollection

    Set pCHPCollection = pTOperator.ConvexHull

    p = 1001

    If pCHPCollection.PointCount = 0 Then Exit Sub

    Set pCHPCollection = pGTOperator.Intersect(pCHPCollection, esriGeometry2Dimension)

    Dim pRPCollection As IPointCollection

    Set pRPCollection = New Ring

    pRPCollection.SetPointCollection pCHPCollection

    pGCollection.AddGeometry pRPCollection

'    Dim pOPolygon As IPointCollection

'    Set pOPolygon = New Polygon

'    pOPolygon.SetPointCollection pPCollection

'

'    Dim pMPCollection As IPointCollection

'    Set pMPCollection = New Multipoint

'    pMPCollection.SetPointCollection pMPCollection

'

'    Dim pTOperator3 As ITopologicalOperator3

'    Set pTOperator3 = pOPolygon

'    Set pOPolygon = pTOperator3.Difference(pMPCollection)

'

'    MsgBox pPCollection.PointCount & " - " & pCHPCollection.PointCount & " = " & pOPolygon.PointCount

'

'    pPCollection.SetPointCollection pOPolygon

    Dim pROperator As IRelationalOperator

    Set pROperator = pCHPCollection

    Set pEVertex = pPCollection.EnumVertices

    Set pPCollection = New Multipoint

    pEVertex.Next pPoint, ltmp, ltmp

    Do Until pPoint Is Nothing

        If pROperator.Contains(pPoint) Then pPCollection.AddPoint pPoint

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

    If pPCollection.PointCount = 0 Then GoTo CEnd

    p = 2000

    Set pTOperator = pPCollection

    Set pCHPCollection = pTOperator.ConvexHull

    p = 2001

    If pCHPCollection.PointCount = 0 Then GoTo CEnd

    Set pCHPCollection = pGTOperator.Intersect(pCHPCollection, esriGeometry2Dimension)

    Set pRPCollection = New Ring

    pRPCollection.SetPointCollection pCHPCollection

    pGCollection.AddGeometry pRPCollection

    AssignZ pGCollection, dAHeight

    aIPolygon.Add pGCollection

Wend

CEnd:

    AssignZ pGCollection, dAHeight

    aIPolygon.Add pGCollection

    Exit Sub

CPError:

    If p = 1000 Then Exit Sub

    If p = 2000 Then Resume Next

    MsgBox "P " & ERR.Description & " " & p

End Sub

'AddZ adds points inside the polygon with high values to the collection

'pPCollection contains the points defining the polygon

Public Sub AddZ()

On Error GoTo cerror:

'    Dim pNCollection As IPointCollection

'    Set pNCollection = New Multipoint

    Set pEVertex = pPCollection.EnumVertices

    Dim pNPoint As IPoint

    pEVertex.Next pPoint, ltmp, ltmp

    Do Until pPoint Is Nothing

        Set pPoint = GetFileCoords(pGRaster, pPoint)

        Set pNPoint = HeighestNeighbor(pPoint)

        Do Until pNPoint Is Nothing

            Set pPoint = pNPoint

            Set pNPoint = HeighestNeighbor(pNPoint)

        Loop

        pPCollection.AddPoint GetUDMCoords(pPoint.x, pPoint.y, pGRaster)

'        pNCollection.AddPoint GetUDMCoords(pPoint.x, pPoint.y, pGRaster)

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

'    Dim pTOperator As ITopologicalOperator

'    Set pTOperator = pNCollection

'

'    Set pNCollection = pTOperator.ConvexHull

'

'    If pNCollection.PointCount = 0 Then Exit Sub

'

'    AssignZ pNCollection, 0

'    aIPolygon.Add pNCollection

    Exit Sub

cerror:

    MsgBox "C " & ERR.Description & " C"

End Sub

'Takes the rings of a polygon and adds them to iPolygon as a polygon

Public Sub AddRings(pPolygon2 As IPolygon2, ByVal dAHeight As Double)

    pPolygon2.SimplifyPreserveFromTo

    Dim lERCount As Long

    Dim lIRCount As Long

    lERCount = pPolygon2.ExteriorRingCount

    ReDim aERing(lERCount)

    pPolygon2.QueryExteriorRingsEx lERCount, aERing(0)

    Dim pERing As IRing

    Dim pRCollection As IPointCollection

    Dim pPCollection As IPointCollection

    Dim iOuter As Integer

    Dim iInner As Integer

    For iOuter = 0 To lERCount - 1

        Set pERing = aERing(iOuter)

        Set pRCollection = pERing

        Set pPCollection = New Polygon

        pPCollection.SetPointCollection pRCollection

        AssignZ pPCollection, dAHeight

        aIPolygon.Add pPCollection

        lIRCount = pPolygon2.InteriorRingCount(pERing)

        ReDim aIRing(lIRCount)

        pPolygon2.QueryInteriorRingsEx pERing, lIRCount, aIRing(0)

        For iInner = 0 To lIRCount - 1

            Set pRCollection = aIRing(iInner)

            Set pPCollection = New Polygon

            pPCollection.SetPointCollection pRCollection

            aIPolygon.Add pPCollection

        Next iInner

    Next iOuter

End Sub

'Returns the point with the highest value that is accessible by

'traveling from pOPoint by always traveling to the highest neighbor

Public Function HeighestNeighbor(pOPoint As IPoint) As IPoint

On Error GoTo h

    Dim x As Integer

    Dim y As Integer

    x = Int(pOPoint.x)

    y = Int(pOPoint.y)

    If aAdded(x, y) Then

        Set HeighestNeighbor = Nothing

        Exit Function

    End If

    Dim pNPoint As IPoint

    Set pNPoint = CreatePoint(pOPoint.x, pOPoint.y)

    If x > 0 Then

        If pGROperator2.Contains(GetUDMCoords(x - 1, y, pGRaster)) Then If aHeight(x - 1, y) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x - 1, y

        If y > 0 Then If pGROperator2.Contains(GetUDMCoords(x - 1, y - 1, pGRaster)) Then If aHeight(x - 1, y - 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x - 1, y - 1

    End If

    If y > 0 Then

        If pGROperator2.Contains(GetUDMCoords(x, y - 1, pGRaster)) Then If aHeight(x, y - 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x, y - 1

        If x < iMWidth - 1 Then If pGROperator2.Contains(GetUDMCoords(x + 1, y - 1, pGRaster)) Then If aHeight(x + 1, y - 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x + 1, y - 1

    End If

    If x < iMWidth - 1 Then

        If pGROperator2.Contains(GetUDMCoords(x + 1, y, pGRaster)) Then If aHeight(x + 1, y) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x + 1, y

        If y < iMHeight - 1 Then If pGROperator2.Contains(GetUDMCoords(x + 1, y + 1, pGRaster)) Then If aHeight(x + 1, y + 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x + 1, y + 1

    End If

    If y < iMHeight - 1 Then

        If pGROperator2.Contains(GetUDMCoords(x, y + 1, pGRaster)) Then If aHeight(x, y + 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x, y + 1

        If x > 0 Then If pGROperator2.Contains(GetUDMCoords(x - 1, y + 1, pGRaster)) Then If aHeight(x - 1, y + 1) > aHeight(pNPoint.x, pNPoint.y) Then pNPoint.PutCoords x - 1, y + 1

    End If

    If pNPoint.x = pOPoint.x And pNPoint.y = pOPoint.y Then Set pNPoint = Nothing

    Set HeighestNeighbor = pNPoint

    Exit Function

h:

    MsgBox "H " & ERR.Description

End Function

'Polify creates a base polygon from which the complete polygon model is made out of

Public Function Polify() As Double

On Error GoTo PlyError:

    Dim pTOperator As ITopologicalOperator

    Dim pCHull As IPointCollection

    Dim pPolygon As IPolygon

    Dim p As Long

    Dim dAHeight As Double

    Dim dBHeight As Double

    dBHeight = 100

    Dim x As Integer

    Dim y As Integer

    Dim EStatus As Integer

    EStatus = 0

 'Points of the same height are grouped together, and a convex hull is created

    For x = 0 To iMWidth - 1

        For y = 0 To iMHeight - 1

            If aVisited(x, y) Then GoTo NxtPly

            If Not pGROperator.Contains(GetUDMCoords(x, y, pGRaster)) Then GoTo NxtPly

            Set aQueue = New esriSystem.Array

            Set pPCollection = New Multipoint

            dTHeight = 0

            p = 0

            Queue x, y

            Do Until p = aQueue.Count

                Set pPoint = aQueue.Element(p)

                Classify pPoint.x, pPoint.y

                p = p + 1

            Loop

            dAHeight = dTHeight / pPCollection.PointCount

            If dAHeight < dBHeight Then dBHeight = dAHeight

            Set pTOperator = pPCollection

            p = 1000

            Set pCHull = pTOperator.ConvexHull

            p = 1001

            If pCHull.PointCount = 0 Then

'                AssignZ pPCollection, dAHeight

'                AddFeature pMPointLayer, pPCollection

            Else

                p = 1002

                Set pPolygon = pGTOperator.Intersect(pCHull, esriGeometry2Dimension)

                If EStatus = 1 Then

                    EStatus = 0

                    GoTo NxtPly

                End If

                p = 1007

                If pPolygon.IsEmpty Then

'                    AssignZ pCHull, dAHeight

'                    aOPolygon.Add pCHull

                Else

                    'Set pPCollection = pPolygon

                    'AddZ

                    p = 1003

                    Set pCHull = pPolygon

                    Set pGROperator2 = pCHull

                    Set pPCollection = New Multipoint

                    pPCollection.SetPointCollection pCHull

                    AddFeature pPolygonLayer, pCHull

                    p = 1004

                    AddZ

                    'AssignZ pPCollection, dAHeight

                    'aIPolygon.Add pPCollection

                    p = 1005

                    'MsgBox "C"

                    Complete_Polygon dAHeight

                    p = 1006

'                    pPolygon.Close

'                    AddRings pPolygon, dAHeight

'                    AssignZ pPolygon, dAHeight

'                    aIPolygon.Add pPolygon

                End If

            End If

NxtPly: Next y

    Next x

    Polify = dBHeight

    Exit Function

PlyError:

    If p = 1000 Then Resume Next

    If p = 1002 Then

        AddFeature pPolygonLayer, pCHull, pCHull.PointCount

        EStatus = 1

        Resume Next

    End If

    MsgBox "A " & ERR.Description & " " & p

End Function

'Initializes variables used in the module and loads arrays with values.

Public Sub DoStuff(pRaster As IRaster, pPolygon As IPolygon)

On Error GoTo yea:

    aHeight = GetPixelArray(pRaster)

    aSlope = GetPixelArray(GetSlope(pRaster))

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    iMWidth = pRProps.Width

    iMHeight = pRProps.Height

    dHError = 0.2 'Val(InputBox("Height Error"))

    ReDim aVisited(iMWidth, iMHeight)

    ReDim aShape(iMWidth, iMHeight)

    ReDim aAdded(iMWidth, iMHeight)

    Set aOPolygon = New esriSystem.Array

    Set aIPolygon = New esriSystem.Array

    AddMPointLayer

    AddMPatchLayer

    AddPointLayer

    AddPolygonLayer

    Set pGRaster = pRaster

    Set pGTOperator = pPolygon

    Set pGROperator = pPolygon

    Dim dBHeight As Double

    dBHeight = Polify

    AddToMPLayer aIPolygon, dBHeight

    Exit Sub

yea:

    MsgBox "Y " & ERR.Description & " " & ERR.Number

End Sub

'Adds all polygons in aPolygon array to the MultiPatch layer

'All polygons are extruded from the base height supplied

Public Sub AddToMPLayer(aPolygon As esriSystem.IArray, ByVal dBHeight As Double)

    Dim pCMPatch As IConstructMultiPatch

    Dim i As Integer

    For i = 0 To aPolygon.Count - 1

        Set pCMPatch = New MultiPatch

        Set pPCollection = aPolygon.Element(i)

        pCMPatch.ConstructExtrudeAbsolute dBHeight, pPCollection

        AddFeature pMPatchLayer, pCMPatch, pPCollection.PointCount

    Next i

End Sub

'Assigns a Z value to all points in the collection

'dAHeight was used to set all points to the same height value, but is no longer used

Public Sub AssignZ(pZPCollection As IPointCollection, ByVal dAHeight As Double)

    Set pZAware = pZPCollection

    pZAware.ZAware = True

    Set pEVertex = pZPCollection.EnumVertices

    pEVertex.Next pPoint, ltmp, ltmp

    Do Until pPoint Is Nothing

        Set pPoint = GetFileCoords(pGRaster, pPoint)

        pEVertex.put_Z aHeight(pPoint.x, pPoint.y)

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

End Sub

'Determines if two heights are equal by checking to see

'if they are within the heigh tolerance of each other

Public Function Equal(ByVal dHeight1 As Double, ByVal dHeight2 As Double) As Boolean

    Equal = dHeight1 - dHError <= dHeight2 And dHeight2 <= dHeight1 + dHError

End Function

'Calls Queue on all the points that are of the same height (as determined by Equal) as the point passed

Public Sub Classify(ByVal x As Integer, ByVal y As Integer)

    Dim dHeight As Double

    dHeight = aHeight(x, y)

    If x > 0 Then

        If Not aVisited(x - 1, y) And Equal(dHeight, aHeight(x - 1, y)) Then Queue x - 1, y

        If y > 0 Then If Not aVisited(x - 1, y - 1) And Equal(dHeight, aHeight(x - 1, y - 1)) Then Queue x - 1, y - 1

    End If

    If y > 0 Then

        If Not aVisited(x, y - 1) And Equal(dHeight, aHeight(x, y - 1)) Then Queue x, y - 1

        If x < iMWidth - 1 Then If Not aVisited(x + 1, y - 1) And Equal(dHeight, aHeight(x + 1, y - 1)) Then Queue x + 1, y - 1

    End If

    If x < iMWidth - 1 Then

        If Not aVisited(x + 1, y) And Equal(dHeight, aHeight(x + 1, y)) Then Queue x + 1, y

        If y < iMHeight - 1 Then If Not aVisited(x + 1, y + 1) And Equal(dHeight, aHeight(x + 1, y + 1)) Then Queue x + 1, y + 1

    End If

    If y < iMHeight - 1 Then

        If Not aVisited(x, y + 1) And Equal(dHeight, aHeight(x, y + 1)) Then Queue x, y + 1

        If x > 0 Then If Not aVisited(x - 1, y + 1) And Equal(dHeight, aHeight(x - 1, y + 1)) Then Queue x - 1, y + 1

    End If

End Sub

'Adds points to the Queue, so that Classify can be called on them

'Also marks the point as visited and adds it to the point collection

'Finally, updates the total height of the polygon

Public Sub Queue(ByVal x As Integer, ByVal y As Integer)

    If pGROperator.Contains(GetUDMCoords(x, y, pGRaster)) Then

        dTHeight = dTHeight + aHeight(x, y)

        pPCollection.AddPoint GetUDMCoords(x, y, pGRaster)

        aVisited(x, y) = True

        aQueue.Add CreatePoint(x, y)

    End If

End Sub

'Iterates over all features in every layer and uses the features as footprints

'Models buildings within the footprints

Public Sub ProcessShapes()

    Dim pELayer As IEnumLayer

    Set pELayer = pSDocument.Scene.Layers

    Dim pLayer As ILayer

    Set pLayer = pELayer.Next

    Dim pRProps As IRasterProps

    Set pRProps = GetRaster()

    Dim pROperator As IRelationalOperator

    Set pROperator = pRProps.Extent

'    AddPointLayer

    Do Until pLayer Is Nothing

        If pLayer.Visible And Not pLayer.Name = "MultiPatch" And Not pLayer.Name = "MultiPoint" And Not pLayer.Name = "Point" Then 'And Not TypeOf pLayer Is RasterLayer Then

            MsgBox "Visible Layer"

            Dim pFLayer As IFeatureLayer

            Set pFLayer = pLayer

            Dim pFClass As IFeatureClass

            Set pFClass = pFLayer.FeatureClass

            Dim pFCursor As IFeatureCursor

            Set pFCursor = pFClass.Search(Nothing, False)

            Dim pFeature As IFeature

            Set pFeature = pFCursor.NextFeature

            Do Until pFeature Is Nothing

                If pROperator.Contains(pFeature.Extent) Then

                    'MsgBox "Start " & pFeature.Value(0)

                    'If pFeature.Value(0) = 146 Then

                    DoStuff Clip(pRProps, pFeature.Extent), pFeature.Shape

'                    AddFeature pPointLayer, pFeature.Extent.LowerLeft

'                    AddFeature pPointLayer, pFeature.Extent.LowerRight

'                    AddFeature pPointLayer, pFeature.Extent.UpperLeft

'                    AddFeature pPointLayer, pFeature.Extent.UpperRight

                    'DoStuff GetRaster(), pFeature.Extent

                End If

                Set pFeature = pFCursor.NextFeature

            Loop

        End If

        Set pLayer = pELayer.Next

    Loop

End Sub

'Module SlopeUse

'Is similar in functionality to HeightUse.  It creates models of buildings,

'however it relies upon slope data, though it still uses height as well

Option Explicit

'Arrays used by the Module

Private aHeight As Variant

Private aSlope As Variant

Private aCurve As Variant

'Maximum Width and Height of the raster

Private iMWidth As Integer

Private iMHeight As Integer

'Global variables for the module

Private pGRaster As IRaster

Private pGPolygon As IPolygon

'Constants defining a low slope and a high slope

Private dHighSlope As Double

Private dLowSlope As Double

'CreateMutliPatch creates and returns a multipatch using the passed geometry

'The geometry is extruded to the height passed

'Even though it accepts any geometry, it has only worked with Polygons so far

Public Function CreateMultiPatch(pGeometry As IGeometry, ByVal dBHeight As Double) As IConstructMultiPatch

    Set CreateMultiPatch = New MultiPatch

    CreateMultiPatch.ConstructExtrudeAbsolute dBHeight, pGeometry

End Function

'AssignZ2 assings Z values to the points in pPCollection

'Heights will be assigned based on the height of the point in the raster

'If dAHeight is not zero, all heights will be set to dAHeight

'If pECollection is passed, and dAHeight is not zero, then the points in

'pECollection will not be assigned dAHeight, but will be assigned the height

'of the corresponding point in the raster

Public Sub AssignZ2(pPCollection As IPointCollection, Optional dAHeight As Double, Optional pECollection As IPointCollection)

On Error GoTo ERR:

    Dim pZAware As IZAware

    Set pZAware = pPCollection

    pZAware.ZAware = True

    Dim pEVertex As IEnumVertex2

    Set pEVertex = pPCollection.EnumVertices

    Dim pPoint As IPoint

    Dim ltmp As Long

    Dim dHeight As Double

    Dim bFile As Boolean

    pEVertex.Next pPoint, ltmp, ltmp

    Do Until pPoint Is Nothing

        bFile = False

        If dAHeight = 0 Then

            bFile = True

        Else

            If Not pECollection Is Nothing Then bFile = Distance(pECollection, pPoint) < 1

        End If

        dHeight = dAHeight

        If bFile Then

            Set pPoint = GetFileCoords(pGRaster, pPoint)

            dHeight = aHeight(pPoint.x, pPoint.y)

        End If

        pEVertex.put_Z dHeight

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

    Exit Sub

ERR:

    MsgBox "C: " & ERR.Description

End Sub

'Returns the average height of the points in pPCollection

Public Function AverageHeight(pPCollection As IPointCollection) As Double

On Error GoTo ERR:

    Dim dTotal As Double

    Dim pEVertex As IEnumVertex

    Set pEVertex = pPCollection.EnumVertices

    Dim pPoint As IPoint

    Dim ltmp As Long

    pEVertex.Next pPoint, ltmp, ltmp

    ltmp = 1

    Do Until pPoint Is Nothing

        Set pPoint = GetFileCoords(pGRaster, pPoint)

        ltmp = 2

        dTotal = dTotal + aHeight(pPoint.x, pPoint.y)

        ltmp = 3

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

    AverageHeight = dTotal / pPCollection.PointCount

    Exit Function

ERR:

    MsgBox "D: " & ERR.Description & " " & ltmp

End Function

'ModelFlat models pPCollection as a Flat roof building

'It does this by assigning z values to the points and creating a multipatch

Public Sub ModelFlat(pPCollection As IPointCollection)

    AssignZ2 pPCollection, AverageHeight(pPCollection)

    AddFeature pMPatchLayer, CreateMultiPatch(pPCollection, 0)

End Sub

'RemoveOutliers removes outlier points in pPCollection

'An Outlier point is defined as being a distance of 7 or more from any other point

Public Sub RemoveOutliers(pPCollection As IPointCollection)

    Dim pEVertex As IEnumVertex

    Set pEVertex = pPCollection.EnumVertices

    Dim pPoint As IPoint

    Dim ltmp As Long

    pEVertex.Next pPoint, ltmp, ltmp

    Do Until pPoint Is Nothing

        pPCollection.RemovePoints ltmp, 1

        'If Distance(pPoint, pPCollection) > 1 Then MsgBox Distance(pPoint, pPCollection)

        If Distance(pPoint, pPCollection) < 7 Then pPCollection.AddPoint pPoint

        pEVertex.Next pPoint, ltmp, ltmp

    Loop

End Sub

'CreateLine creates and returns a line using the first two points in pICollection

Public Function CreateLine(pICollection As IPointCollection) As ILine

    Set CreateLine = New esriGeometry.Line

    Dim pPoint As IPoint

    Set pPoint = New Point

    pICollection.QueryPoint 0, pPoint

    CreateLine.ToPoint = pPoint

    pICollection.QueryPoint 1, pPoint

    CreateLine.FromPoint = pPoint

End Function

'Tolerates determines if the firs two parameters are within dTolerance of each other

Public Function Tolerate(dLow As Double, dHigh As Double, dTolerance As Double) As Boolean

    'Tolerate = dLow - dTolerance <= dHigh And dHigh <= dLow + dTolerance

    Tolerate = Abs(dHigh - dLow) <= dTolerance

End Function

'Makes the passed geometry ZAware

Public Sub MakeZAware(pZAware As IZAware)

    pZAware.ZAware = True

End Sub

'CreatePolygonModel creates a polygon model using pSCollectino and pICollection

'Only the first two points of pICollection are used

'For each segment in pSCollection, a triangle is made using the closer of the points in pICollection

'or a quadrilateral is made using both points in pICollection

Public Sub CreatePolygonModel(pSCollection As ISegmentCollection, pICollection As IPointCollection)

On Error GoTo ERR_

    Dim dAngle As Double

    dAngle = CreateLine(pICollection).Angle

    Dim pESegment As IEnumSegment

    Set pESegment = pSCollection.EnumSegments

    Dim ltmp As Long

    Dim pLine As ILine

    pESegment.Next pLine, ltmp, ltmp

    Dim pi As Double

    pi = 4 * Atn(1)

    Do Until pLine Is Nothing

        Dim pPCollection As IPointCollection

        If Tolerate((dAngle + pi) Mod pi, (pLine.Angle + pi) Mod pi, pi / 36) Then

            Set pPCollection = New Multipoint

            pPCollection.SetPointCollection pICollection

            pPCollection.AddPoint pLine.FromPoint

            pPCollection.AddPoint pLine.ToPoint

            Set pPCollection = ConvexHull(pPCollection)

            AssignZ2 pPCollection, AverageHeight(pSCollection), pICollection

            AddFeature pMPatchLayer, CreateMultiPatch(pPCollection, 0)

        Else

            Set pPCollection = New Polygon

            MakeZAware pPCollection

            pPCollection.AddPoint pLine.FromPoint

            pPCollection.AddPoint pLine.ToPoint

            Dim pPoint1 As IPoint

            Set pPoint1 = New Point

            pICollection.QueryPoint 0, pPoint1

            Dim pPoint2 As IPoint

            Set pPoint2 = New Point

            pICollection.QueryPoint 1, pPoint2

            If Distance(pLine, pPoint1) < Distance(pLine, pPoint2) Then Set pPoint2 = pPoint1

            pPCollection.AddPoint pPoint2

            AddFeature pMPatchLayer, CreateMultiPatch(pPCollection, 0)

        End If

        pESegment.Next pLine, ltmp, ltmp

    Loop

    Exit Sub

ERR_:

    MsgBox "Z: " & ERR.Description & " " & ltmp

End Sub

'ModelPeaked models pOCollection and pICollecion as a peaked building

'pOCollection is a collection of points defining the outline of the building

'pICOllection is a collection of interior points defining the peak

Public Sub ModelPeaked(pOCollection As IPointCollection, pICollection As IPointCollection)

    RemoveOutliers pICollection

    Set pICollection = FurthestPoints(pICollection)

    AssignZ2 pOCollection, AverageHeight(pOCollection)

    AssignZ2 pICollection

    CreatePolygonModel pOCollection, pICollection

End Sub

'Distance calculates and returns the distance between the two geometries

Public Function Distance(pGeometry1 As IGeometry, pGeometry2 As IGeometry) As Double

    Dim pPOperator As IProximityOperator

    Set pPOperator = pGeometry1

    Distance = pPOperator.ReturnDistance(pGeometry2)

'    Distance2 = (pPoint2.x - pPoint1.x) * (pPoint2.x - pPoint1.x) + (pPoint2.y - pPoint1.y) * (pPoint2.y - pPoint2.y)

End Function

'FurtheestPoints determines the two points furthest from each other in pPCollection

'These two points are returned in a point collection

Public Function FurthestPoints(pPCollection As IPointCollection) As IPointCollection

    Dim pEVertex1 As IEnumVertex

    Set pEVertex1 = pPCollection.EnumVertices

    Dim ltmp As Long

    Dim dLongest As Double

    Dim pKPoint1 As IPoint

    Dim pKPoint2 As IPoint

    Dim pPoint1 As IPoint

    pEVertex1.Next pPoint1, ltmp, ltmp

    Do Until pPoint1 Is Nothing

        Dim pEVertex2 As IEnumVertex

        Set pEVertex2 = pPCollection.EnumVertices

        Dim pPoint2 As IPoint

        pEVertex2.Next pPoint2, ltmp, ltmp

        Do Until pPoint2 Is Nothing

            If Distance(pPoint1, pPoint2) > dLongest Then

                dLongest = Distance(pPoint1, pPoint2)

                Set pKPoint1 = pPoint1

                Set pKPoint2 = pPoint2

            End If

            pEVertex2.Next pPoint2, ltmp, ltmp

        Loop

        pEVertex1.Next pPoint1, ltmp, ltmp

    Loop

    Set FurthestPoints = New Polygon

    FurthestPoints.AddPoint pKPoint1

    FurthestPoints.AddPoint pKPoint2

End Function

'HighSlope determines if the slope between two points is considered to be a high slope

Public Function HighSlope(x1, y1, x2, y2) As Boolean

    HighSlope = 180 * Atn(Abs(aHeight(x2, y2) - aHeight(x1, y1))) / (4 * Atn(1)) >= dHighSlope

End Function

'Valid determines if the coordinate passed is valid for the raster

Public Function Valid(x, y) As Boolean

    Valid = True

    If x <= 0 Then Valid = False

    If y <= 0 Then Valid = False

    If x >= iMWidth Then Valid = False

    If y >= iMHeight Then Valid = False

End Function

'Heighest determines if the point passed is higher than the neighboring points

'Only points that the passed point has a slope with are compared

Public Function Heighest(x, y) As Boolean

On Error GoTo h

    Heighest = True

    Dim dHeight As Double

    dHeight = aHeight(x, y)

    Dim Count As Integer

    Count = 0

    If Valid(x - 1, y - 1) Then

        If HighSlope(x, y, x - 1, y - 1) Then Count = Count + 1

        If HighSlope(x, y, x - 1, y - 1) And aHeight(x - 1, y - 1) > dHeight Then Heighest = False

    End If

    If Valid(x - 1, y) Then

        If HighSlope(x, y, x - 1, y) Then Count = Count + 1

        If HighSlope(x, y, x - 1, y) And aHeight(x - 1, y) > dHeight Then Heighest = False

    End If

    If Valid(x - 1, y + 1) Then

        If HighSlope(x, y, x - 1, y + 1) Then Count = Count + 1

        If HighSlope(x, y, x - 1, y + 1) And aHeight(x - 1, y + 1) > dHeight Then Heighest = False

    End If

    If Valid(x, y - 1) Then

        If HighSlope(x, y, x, y - 1) Then Count = Count + 1

        If HighSlope(x, y, x, y - 1) And aHeight(x, y - 1) > dHeight Then Heighest = False

    End If

    If Valid(x, y + 1) Then

        If HighSlope(x, y, x, y + 1) Then Count = Count + 1

        If HighSlope(x, y, x, y + 1) And aHeight(x, y + 1) > dHeight Then Heighest = False

    End If

    If Valid(x + 1, y - 1) Then

        If HighSlope(x, y, x + 1, y - 1) Then Count = Count + 1

        If HighSlope(x, y, x + 1, y - 1) And aHeight(x + 1, y - 1) > dHeight Then Heighest = False

    End If

    If Valid(x + 1, y) Then

        If HighSlope(x, y, x + 1, y) Then Count = Count + 1

        If HighSlope(x, y, x + 1, y) And aHeight(x + 1, y) > dHeight Then Heighest = False

    End If

    If Valid(x + 1, y + 1) Then

        If HighSlope(x, y, x + 1, y + 1) Then Count = Count + 1

        If HighSlope(x, y, x + 1, y + 1) And aHeight(x + 1, y + 1) > dHeight Then Heighest = False

    End If

    'MsgBox Count

    Exit Function

h:

    MsgBox "H " & ERR.Description

End Function

'ConvexHull creates and returns the convexhull of the pointcollection passed

Public Function ConvexHull(pPCollection As IPointCollection) As IPolygon

    Dim pTOperator As ITopologicalOperator

    Set pTOperator = pPCollection

    Set ConvexHull = pTOperator.ConvexHull

End Function

'Buffer creates and returns pPolygon buffered by iDistance

Public Function Buffer(pPolygon As IPolygon, iDistance As Integer) As IPolygon

    Dim pTOperator As ITopologicalOperator

    Set pTOperator = pPolygon

    Set Buffer = pTOperator.Buffer(iDistance)

End Function

'PointClassifybydSlope attempts to model buildings and uses slope values to classify points

'Points with a curve > 85 and a  high slope are added to the outline point collection

'Points with a low slope are added to the interior point collection

'Based on the number of interior points, the two collections are used to either create

'Either a Flat model or a Peacked Model

Public Sub PointClassifybydSlope()

On Error GoTo ERR:

    Dim x As Integer

    Dim y As Integer

    Dim iTotal As Integer

    iTotal = 0

    Dim pROperator As IRelationalOperator

    Set pROperator = Buffer(pGPolygon, 2)

    Dim pROperator2 As IRelationalOperator

    Set pROperator2 = pGPolygon

    Dim pPCollection As IPointCollection

    Set pPCollection = New Multipoint

    Dim pRPCollection As IPointCollection

    Set pRPCollection = New Multipoint

    For x = 0 To iMWidth - 1

        For y = 0 To iMHeight - 1

            If pROperator.Contains(GetUDMCoords(x, y, pGRaster)) Then

                If aCurve(x, y) >= 85 And aSlope(x, y) >= dHighSlope Then

                    If Heighest(x, y) Then pRPCollection.AddPoint GetUDMCoords(x, y, pGRaster)

                End If

                If pROperator2.Contains(GetUDMCoords(x, y, pGRaster)) Then

                    iTotal = iTotal + 1

                    If aSlope(x, y) <= dLowSlope Then 'And aCurve(x, y) >= 70

                        pPCollection.AddPoint GetUDMCoords(x, y, pGRaster)

                    End If

                End If

            End If

        Next y

    Next x

    AddFeature pMPointLayer, pPCollection

    AddFeature pMPointLayer, pRPCollection

    Set pRPCollection = ConvexHull(pRPCollection)

'    AddFeature pPolygonLayer, pRPCollection

'    Dim pPLCollection As IPointCollection

'    Set pPLCollection = New Polyline

'    pPLCollection.SetPointCollection pRPCollection

'    AddFeature pPolylineLayer, pPLCollection

'    Dim pPolycurve As IPolycurve

'    Set pPolycurve = pPLCollection

'    AddFeature pPointLayer, pPolycurve.FromPoint

'    AddFeature pPointLayer, pPolycurve.ToPoint

'    Dim pPOperator As IProximityOperator

'    Set pPOperator = pPolycurve

'    pPolycurve.ToPoint = pPolycurve.FromPoint

'    AddFeature pPointLayer, pPolycurve.ToPoint

    'AddFeature pPolylineLayer, pPolycurve

    If pPCollection.PointCount < 0.25 * iTotal Then

        'AddFeature pMPointLayer, pPCollection

        ModelPeaked pRPCollection, pPCollection

    End If

'    MsgBox pPCollection.PointCount & " " & iTotal

    If pPCollection.PointCount > 0.65 * iTotal Then

        ModelFlat pRPCollection

        'AddFeature pPolygonLayer, pTOperator

    End If

'    Set pRPCollection = pTOperator

'    Set pPLCollection = New Polyline

'    pPLCollection.SetPointCollection pRPCollection

'    AddFeature pPolylineLayer, pPLCollection

'    Set pPolycurve = pPLCollection

'    pPolycurve.Weed 0.001

'    AddFeature pPolylineLayer, pPLCollection

'    Dim pTPCollection As IPointCollection

'    Set pTPCollection = New Triangles

'    pTPCollection.SetPointCollection pRPCollection

'    AssignZ2 pRPCollection, 0

'    Set pTPCollection = New Triangles

'    pTPCollection.SetPointCollection pRPCollection

'    Dim pGCollection As IGeometryCollection

'    Set pGCollection = New MultiPatch

'    pGCollection.AddGeometry pTPCollection

'    AddFeature pMPatchLayer, pGCollection

    Exit Sub

ERR:

    MsgBox ERR.Description

End Sub

'ToMultiPoint creates and returns the given pPCollection as a multipoint

Public Function ToMultiPoint(pPCollection As IPointCollection) As IPointCollection

    Set ToMultiPoint = New Multipoint

    ToMultiPoint.SetPointCollection pPCollection

End Function

'SlopeModel sets up variables to be used by this module, and then calls PointClassifybydSlope

Public Sub SlopeModel(pRaster As IRaster, pPolygon As IPolygon)

On Error GoTo ERR:

    Dim pSRaster As IRaster

    Set pSRaster = GetSlope(pRaster)

    aHeight = GetPixelArray(pRaster)

    aSlope = GetPixelArray(pSRaster)

    aCurve = GetPixelArray(GetSlope(pSRaster))

    Dim pRProps As IRasterProps

    Set pRProps = pRaster

    iMWidth = pRProps.Width

    iMHeight = pRProps.Height

    dHighSlope = 47.04085182

    dLowSlope = 12.54422715

    AddMPointLayer

    AddMPatchLayer

    AddPointLayer

    AddPolygonLayer

    AddPolylineLayer

    Set pGRaster = pRaster

    Set pGPolygon = pPolygon

'    MsgBox "Start"

    PointClassifybydSlope

    Exit Sub

ERR:

    MsgBox "A: " & ERR.Description

End Sub

'ProcessSlope iterates over the features in every visible layer

'Each feature that is contained within the raster is used as an outline for creating a model

Public Sub ProcessSlope()

    Dim pELayer As IEnumLayer

    Set pELayer = pSDocument.Scene.Layers

    Dim pLayer As ILayer

    Set pLayer = pELayer.Next

    Dim pRProps As IRasterProps

    Set pRProps = GetRaster()

    Dim pROperator As IRelationalOperator

    Set pROperator = pRProps.Extent

    Do Until pLayer Is Nothing

        If pLayer.Visible And Not pLayer.Name = "MultiPatch" And Not pLayer.Name = "MultiPoint" And Not pLayer.Name = "Point" Then 'And Not TypeOf pLayer Is RasterLayer Then

            Dim pFLayer As IFeatureLayer

            Set pFLayer = pLayer

            Dim pFClass As IFeatureClass

            Set pFClass = pFLayer.FeatureClass

            Dim pFCursor As IFeatureCursor

            Set pFCursor = pFClass.Search(Nothing, False)

            Dim pFeature As IFeature

            Set pFeature = pFCursor.NextFeature

            Do Until pFeature Is Nothing

                If pROperator.Contains(pFeature.Extent) Then SlopeModel Clip(pRProps, pFeature.Extent), pFeature.Shape

                Set pFeature = pFCursor.NextFeature

            Loop

        End If

        Set pLayer = pELayer.Next

    Loop

End Sub

'Class Module Generate2DPrints

'Allows the user to select 4 points and define a footprint

'Footprint passed to SlopeModel in Module SlopeUse

Option Explicit

Implements ICommand

Implements ITool

Private pRaster As IRaster

Private pPCollection As IPointCollection

Private Property Get ICommand_Bitmap() As esriSystem.OLE_HANDLE

    ICommand_Bitmap = frmPics.cmdCompass.Picture.Handle

End Property

Private Property Get ICommand_Caption() As String

    ICommand_Caption = "Generate 2D Footprints"

End Property

Private Property Get ICommand_Category() As String

    ICommand_Category = "Urban Modeler Tools"

End Property

Private Property Get ICommand_Checked() As Boolean

End Property

Private Property Get ICommand_Enabled() As Boolean

    ICommand_Enabled = TypeOf pSDocument.SelectedLayer Is RasterLayer

End Property

Private Property Get ICommand_HelpContextID() As Long

End Property

Private Property Get ICommand_HelpFile() As String

End Property

Private Property Get ICommand_Message() As String

    ICommand_Message = "Generates 2D Footprints for the selected raster layer"

End Property

Private Property Get ICommand_Name() As String

    ICommand_Name = "UrbanModeling_2DFootprints"

End Property

Private Sub ICommand_OnClick()

End Sub

Private Sub ICommand_OnCreate(ByVal hook As Object)

    Init hook

End Sub

Private Property Get ICommand_Tooltip() As String

    ICommand_Tooltip = "Generate 2D Footprints"

End Property

Private Property Get ITool_Cursor() As esriSystem.OLE_HANDLE

End Property

Private Function ITool_Deactivate() As Boolean

    ITool_Deactivate = True

End Function

Private Function ITool_OnContextMenu(ByVal x As Long, ByVal y As Long) As Boolean

End Function

Private Sub ITool_OnDblClick()

End Sub

Private Sub ITool_OnKeyDown(ByVal keyCode As Long, ByVal shift As Long)

End Sub

Private Sub ITool_OnKeyUp(ByVal keyCode As Long, ByVal shift As Long)

End Sub

Private Sub ITool_OnMouseMove(ByVal button As Long, ByVal shift As Long, ByVal x As Long, ByVal y As Long)

End Sub

Private Sub ITool_OnMouseUp(ByVal button As Long, ByVal shift As Long, ByVal x As Long, ByVal y As Long)

End Sub

Private Sub ITool_Refresh(ByVal hdc As esriSystem.OLE_HANDLE)

End Sub

Private Sub ITool_OnMouseDown(ByVal button As Long, ByVal shift As Long, ByVal x As Long, ByVal y As Long)

    If pRaster Is Nothing Then Set pRaster = GetRaster()

    If pPCollection Is Nothing Then Set pPCollection = New Polygon

    pPCollection.AddPoint GetUMDCoords(x, y)

    If pPCollection.PointCount = 4 Then

        Dim pPolygon As IPolygon

        Set pPolygon = pPCollection

        pPolygon.Close

        SlopeModel Clip(pRaster, pPolygon.Envelope), pPolygon

        Set pPCollection = New Polygon

    End If

End Sub

'Class Module Process2DPrints

'Uses the 2D footprints to create models of the buildings

'Uses Module HeightUse and Function ProccessShapes

Option Explicit

Implements ICommand

Private Property Get ICommand_Bitmap() As esriSystem.OLE_HANDLE

    ICommand_Bitmap = frmPics.cmdWrench.Picture.Handle

End Property

Private Property Get ICommand_Caption() As String

    ICommand_Caption = "Process 2D Blueprints"

End Property

Private Property Get ICommand_Category() As String

    ICommand_Category = "Urban Modeler Tools"

End Property

Private Property Get ICommand_Checked() As Boolean

End Property

Private Property Get ICommand_Enabled() As Boolean

    ICommand_Enabled = TypeOf pSDocument.SelectedLayer Is RasterLayer

End Property

Private Property Get ICommand_HelpContextID() As Long

End Property

Private Property Get ICommand_HelpFile() As String

End Property

Private Property Get ICommand_Message() As String

End Property

Private Property Get ICommand_Name() As String

    ICommand_Name = "2D Blueprints Processor"

End Property

Private Sub ICommand_OnClick()

    ProcessShapes

End Sub

Private Sub ICommand_OnCreate(ByVal hook As Object)

    Init hook

End Sub

Private Property Get ICommand_Tooltip() As String

    ICommand_Tooltip = "Process 2D Blueprints"

End Property

'Class Module ProcessSelected

'Uses 2d Footprints to create models

'Uses Module SlopeUse and Function ProcessSlope

Option Explicit

Implements ICommand

Private Property Get ICommand_Bitmap() As esriSystem.OLE_HANDLE

    ICommand_Bitmap = frmPics.cmdWrench.Picture.Handle

End Property

Private Property Get ICommand_Caption() As String

    ICommand_Caption = "Process Blueprints"

End Property

Private Property Get ICommand_Category() As String

    ICommand_Category = "Urban Modeler Tools"

End Property

Private Property Get ICommand_Checked() As Boolean

End Property

Private Property Get ICommand_Enabled() As Boolean

    ICommand_Enabled = TypeOf pSDocument.SelectedLayer Is RasterLayer

End Property

Private Property Get ICommand_HelpContextID() As Long

End Property

Private Property Get ICommand_HelpFile() As String

End Property

Private Property Get ICommand_Message() As String

End Property

Private Property Get ICommand_Name() As String

    ICommand_Name = "Blueprints Processor"

End Property

Private Sub ICommand_OnClick()

    ProcessSlope

End Sub

Private Sub ICommand_OnCreate(ByVal hook As Object)

    Init hook

End Sub

Private Property Get ICommand_Tooltip() As String

    ICommand_Tooltip = "Process Blueprints"

End Property

Figure 1: Extracted building model without extraneous area removed





Figure 2: Extracted building model with extraneous area removed





Figure 3: Extracted Building Model with descriptive interior points





Figure 4: First attempt to fill interior hole





Figure 5: Final Building Model – Hole completely filled and artifacts removed.








